

Lord Tactile Sensor Driver Program

For LTS 2001 Tactile Sensor With Rev 1.162 Software

By: Dan Kohn and Alf Bowser

For: Clark Pick

Dec. 5, 1986

## Running the Lord Sensor Demo Program

This program demonstrates the uses of a Tactile Sensor. To run the program make sure the sensor is connected to the properly to the power supply and computer. Also make sure that the sensor pad is clear for installation. To run the program put the disk marked LORD SENSOR DEMO PROGRAM into the A: drive of an IBM XT with and RGB monitor and an installed Com1: port. Then hold down the Ctrl-Alt-Del keys simultaneously (reboot). This will automatically remind you to turn the power to the sensor on and, after you hit any key, will load in the program. The first screen that comes on will tell you that it is initializing the sensor. After initialization is complete it will ask you to hit any key. At this time if you want to get further instructions type the letter I. This will give you a list of commands and how to use them. If you hit any other key, or after you are done reading the help screens, the Menu will appear. From this menu you can access 6 commands. They are:

1) Get data from the sensor and save it on disk.

This allows you to get data from the sensor and save it for later use. It will also show you a view of the part with a color scale representing the amount of force applied to the sensor at each point. Make sure you have a part on the sensor before running this command.

2) Show 3 dimensional view of data on disk.

For this command you must have run command 1 first. After you have a part saved to disk this command will draw a 3 dimensional view of the surface of the part that is touching the sensor pad.

3) Compare part on disk to new part.

For this command you must have run command 1 first. After you have save a part to disk this command will compare the new part to the one on the disk. Once the command is picked it will ask you for a tolerance (between 0 and 100). Make sure you have a part on the pad before hitting return. After you run this command the same screen that appeared on command 1 will appear again, but after the scan is completed information will appear at the right hand side of the screen informing you of the area of the old part, area of the new part and weather or not the part passed (within the tolerance you picked).

Note: The total area of your part makes a

difference on the outcome. If your good part covers the entire sensor pad and the next part is missing one pixel and you use a 10% tolerance, the part will be accepted because  $160 \pm 10\%$  gives a range of 144 to 176, and 159 (full sensor-1) is within that range. But on the other hand if you have a good part with 10 pixels on the pad covered and a tolerance of 10% the range of the part it will except is 9 to 11 pixels. Therefore this algorithm is more accurate for full scan parts.

#### 4) Edge detection.

Before running this command make sure the part you want to use is already on the pad. Once the command is entered the screen that is used in command 1 is shown. After the sensor sends all the data the computer will show an angle of rotation and an offset to the center of the part that a robot would have to move to place the part in the desired location. This program assumes you would want the part in the exact center of the pad.

Notes: The part must cover the pad from one side to the other. Also the part must be placed on the pad from row zero up for this algorithm to work. Only one direction and the angle was given because the other direction would be a part dependent calculation.

#### 5) Help.

Shows this information.

#### 6) End program.

Stops execution of the program.

#### General notes:

The sensor must be on when the program is executed and must remain on until you are done with the program (you cannot turn the sensor on and off during program operation).

You can use the print screen key to get a printout of the 3 dimensional view.

### Basic Program Notes

<u>Line #</u>	<u>Comments</u>
1000	On any error call the error trapping subroutine
1020-1160	Print opening screen
1190	Dimension all Arrays
1200-1220	read in color scale data (at line 3090)
1230	Open communication port (Com1:) for use at 9600 baud, no parity, 8 data bits, 2 stop bits.
1240	Command sensor to send an initial scan (all data from sensor should be 0)
1250-1290	Read in data from sensor and discard it
1320-1360	Inform user that initialization is done and ask if they want more information on the program
1390-1460	Print out command menu
1470-1490	Get command choice and make sure it is valid
1500	go to subroutines for different commands
1510	reprint the menu
1550-1640	print border and color scale for sensor data displays
1650	Open disk file if command #1 was chosen from menu. The new data will be saved over any existing file named "Lord.Dat".
1680	Command to take a scan of the sensor pad.
1690-1800	Gets information from the sensor and changes the information into number that are stored in a two dimensional array. If command 1 was the last chosen the data will also be saved to disk.
1810	Close data file if opened
1820-1840	Wait until user is done looking at data.

1850	return to origin
1880	run the read data from disk subroutine
1900-1910	go into hi-res graphics and clear the screen
1930-2040	plot x,y,z co-ordinates and record them as 2 dimensional data
2070-2120	clear screen and replot data as a x-direction grid.
2150-2210	overlay a y-direction grid over the x-grid
2220-2240	wait for user to hit any key and return to line 1510
2270-2350	Read data file (created by command #1) and return
2380	Clear variables area(1) and area(2)
2390	read data from disk
2400-2410	get tolerance and change it to percent
2420-2460	copy array P into array C
2470	goto subroutine to get data from sensor and display it
2480-2530	calculate area of both the part on disk and the one just read from the sensor (each pixel that is on has an area of $.071^2$ )
2540	compare new part to old part with tolerances added
2550-2580	print results of test
2590-2610	wait for user input then return to menu
2640	clear Y array
2650	goto subroutine to get data from sensor and display it
2660-2710	find the last pixel in columns 1,5,6 and 10 and record their location.

2720	check to see if angle of rotation is 90 degrees if so goto line 2740
2730-2760	calculate the angle of rotation (and convert it to degrees)
2770	print angle of rotation
2780-2790	calculate the distance from center of the sensor the part
2800	Print the distance
2810-2830	waits for user input then return
2860-3020	read help file from disk and display it
3050-3080	Error trapping (only traps i/o errors at lines 1250 and 1270)
3090	data for color scale

#### Comments

The program can be modified so that the results of the edge detection and the parts comparison can be sent to a robot. To do this you must have two communications ports on the IBM. Any information you want the robot to receive could be sent by way of Com2:.

The program was written in GWBASIC version 3.2 and then compiled in Microsoft Quick Basic Compiler 1.02.

## Program Listing

```

1000 ON ERROR GOTO 3040
1010 :
1020 REM ***** OPENING SCREEN *****
1030 SCREEN 0
1040 CLS
1050 COLOR 6
1060 PRINT STRING$(79,219)
1070 FOR N=1 TO 21
1080     PRINT STRING$(2,219);STRING$(75,32);STRING$(2,219)
1090 NEXT N
1100 PRINT STRING$(79,219)
1110 LOCATE 6,30:PRINT "LORD TACTILE SENSOR"
1120 LOCATE 8,32:PRINT "DRIVER PROGRAM"
1130 LOCATE 10,15:PRINT "FOR LTS 200I TACTILE SENSOR WITH REV 1.162 SOFTWARE"
1140 LOCATE 12,30:PRINT "BY:  DAN KOHN"
1150 LOCATE 13,30:PRINT "      ALF BOWSER"
1160 LOCATE 18,30:COLOR 22:PRINT "INITIALIZING SENSOR"
1170 :
1180 REM ***** INITIALIZE PROGRAM AND SENSOR *****
1190 DIM P(11,17),C(11,17),L(11,17),AREA(2),Y(4),CS(15)
1200 FOR T=0 TO 15
1210     READ CS(T)
1220 NEXT T
1230 OPEN "com1:9600,n,8,2" AS #1
1240 PRINT #1,CHR$(27)+"[?1;1L"
1250 FOR Y=1 TO 16
1260     FOR X=1 TO 10
1270         A$=INPUT$(1,1)
1280     NEXT X
1290 NEXT Y
1300 :
1310 REM ***** INSTRUCTIONS *****
1320 LOCATE 18,30:COLOR 6:PRINT "INITIALIZATION COMPLETED"
1330 LOCATE 20,15:PRINT "PRESS SPACE TO CONTINUE OR I FOR MORE INFORMATION"
1340 Q$=INKEY$
1350 IF Q$="" THEN GOTO 1340
1360 IF Q$="i" OR Q$="I" THEN GOSUB 2850
1370 :
1380 REM ***** MENU *****
1390 SCREEN 0:CLS:COLOR 7
1400 LOCATE 5,20:PRINT "Program for the Lord Tactile Sensor"
1410 LOCATE 7,15:PRINT "1) Get data from sensor and save it on disk"
1420 LOCATE 9,15:PRINT "2) Show 3 dimensional view of data on disk"
1430 LOCATE 11,15:PRINT "3) Compare part on disk to new part"
1440 LOCATE 13,15:PRINT "4) Edge detection"
1450 LOCATE 15,15:PRINT "5) Help"
1460 LOCATE 17,15:PRINT "6) End program"
1470 LOCATE 19,27:INPUT "choice";Q$
1480 LOCATE 21,18:PRINT SPACE$(31)
1490 IF Q$>CHR$(48) AND Q$<CHR$(55) THEN Q=VAL(Q$) ELSE BEEP:LOCATE 21,18:PRINT
"illegal Entry, Please try again":LOCATE 19,35:PRINT SPACE$(LEN(Q$)):GOTO 1470
1500 ON Q GOSUB 1540,1870,2370,2630,2850,1520
1510 GOTO 1370
1520 END
1530 :
1540 REM **** SCREEN SET UP ****
1550 CLS:SCREEN 0
1560 PRINT CHR$(218);STRING$(20,196);CHR$(191)
1570 FOR N=1 TO 16
1580     PRINT CHR$(179);STRING$(20,32);CHR$(179);" ";
1590     COLOR CS(N-1),0,0
1600     PRINT CHR$(219);CHR$(219);" ";

```



```

1620 PRINT N-1
1630 NEXT N
1640 PRINT CHR$(192);STRING$(20,196);CHR$(217)
1650 IF Q=1 THEN OPEN "O",#2,"LORD.DAT"
1660 :
1670 REM ***** GET INFO FROM SENSOR *****
1680 PRINT#1,CHR$(27)+"[?1;1L"
1690 FOR Y=1 TO 16
1700 LOCATE Y+1,2
1710 FOR X=1 TO 10
1720 A$=INPUT$(1,1)
1730 P=INT(((ASC(A$))/2)+.5)
1740 P(X,Y)=ASC(A$)
1750 IF Q=1 THEN PRINT #2,P(X,Y)
1760 COLOR CS(P),0,0
1770 PRINT STRING$(2,219);
1780 IF X<>10 THEN COLOR 7,0,0:PRINT STRING$(2,196);STRING$(2,29);
1790 NEXT X
1800 NEXT Y
1810 IF Q=1 THEN CLOSE #2
1820 COLOR 7,0,0
1830 IF Q=1 THEN LOCATE 22,2:PRINT "PRESS RETURN TO CONTINUE" ELSE GOTO 1850
1840 IF INKEY$="" THEN GOTO 1840
1850 RETURN
1860 :
1870 REM ***** 3-D PLOT OF DATA *****
1880 GOSUB 2260
1890 REM ***** 1st PLOT (GET X,Y,Z X CO-ORDINATES) *****
1900 SCREEN 2
1910 CLS
1920 DRAW "BF65"
1930 FOR Y=0 TO 17
1940 DRAW "BTA64 BU9"
1950 FOR X=0 TO 11
1960 IF Y/2<>INT(Y/2) THEN J=11-X ELSE J=X
1970 Z=P(J,Y)
1980 IF Y/2<>INT(Y/2) AND X>0 THEN DRAW "BTA116 BU11" ELSE IF Y/2=INT(Y/2) A
ND X>0 THEN DRAW "BTA296 BU11"
1990 DRAW "TA0 U="+VARPTR$(Z)
2000 C(X,Y)=POINT(0)
2010 L(X,Y)=POINT(1)
2020 DRAW "BTA180 BU="+VARPTR$(Z)
2030 NEXT X
2040 NEXT Y
2050 :
2060 REM ***** 2nd PLOT (X GOING GRID) *****
2070 CLS:DRAW "BF65"
2080 FOR Y=0 TO 17
2090 FOR X=0 TO 11
2100 IF X=0 AND Y=0 THEN DRAW "BM="+VARPTR$(C(X,Y))+",="+VARPTR$(L(X,Y)) EL
E DRAW "M="+VARPTR$(C(X,Y))+",="+VARPTR$(L(X,Y))
2110 NEXT X
2120 NEXT Y
2130 :
2140 REM ***** 3rd PLOT (Y GOING GRID) *****
2150 FOR X=0 TO 11
2160 FOR Y=0 TO 17
2170 IF Y/2<>INT(Y/2) THEN J=11-X ELSE J=X
2180 DRAW "M="+VARPTR$(C(J,Y))+",="+VARPTR$(L(J,Y))
2190 NEXT Y
2200 IF X<>11 THEN DRAW "BM="+VARPTR$(C(X,0))+",="+VARPTR$(L(X,0))
2210 NEXT X
2220 PRINT "PRESS RETURN TO CONT"
2230 IF INKEY$="" THEN GOTO 2230
2240 RETURN

```

```

2260 REM **** READ DATA FROM DISK ****
2270 OPEN "I",#2,"LORD.DAT"
2280 FOR Y=1 TO 16
2290     FOR X=1 TO 10
2300         LINE INPUT #2,Q$
2310         P(X,Y)=VAL(Q$)
2320     NEXT X
2330 NEXT Y
2340 CLOSE #2
2350 RETURN
2360 :
2370 REM ***** AREA COMPARASON *****
2380 AREA(1)=0:AREA(2)=0
2390 GOSUB 2260
2400 INPUT "TOLERANCE VALUE (IN %) ";TOL
2410 TOL=TOL/100
2420 FOR Y=1 TO 16
2430     FOR X=1 TO 10
2440         C(X,Y)=P(X,Y)
2450     NEXT X
2460 NEXT Y
2470 GOSUB 1540
2480 FOR Y=1 TO 16
2490     FOR X=1 TO 10
2500         IF C(X,Y)>0 THEN AREA(1)=AREA(1)+.071^2
2510         IF P(X,Y)>0 THEN AREA(2)=AREA(2)+.071^2
2520     NEXT X
2530 NEXT Y
2540 IF AREA(2)>AREA(1)-AREA(1)*TOL AND AREA(2)<AREA(1)+AREA(1)*TOL THEN CHECK$=
"WITHIN TOLERANCE" ELSE CHECK$="NOT WITHIN TOLERANCE"
2550 LOCATE 5,40:PRINT "AREA OF GOOD PART:";AREA(1)
2560 LOCATE 7,40:PRINT "AREA OF PART TO BE CHECKED:";AREA(2)
2570 LOCATE 9,40:PRINT "TOLERANCE:";TOL
2580 LOCATE 11,40:PRINT "THE PART IS ";CHECK$
2590 LOCATE 22,2:PRINT "PRESS RETURN TO CONTINUE"
2600 IF INKEY$="" THEN GOTO 2600
2610 RETURN
2620 :
2630 REM **** EDGE DETECTION ****
2640 FOR N=1 TO 4:Y(N)=0:NEXT N
2650 GOSUB 1540
2660 FOR N=1 TO 16
2670     IF P(1,N)>4 THEN Y(1)=N
2680     IF P(10,N)>4 THEN Y(2)=N
2690     IF P(5,N)>4 THEN Y(3)=N
2700     IF P(6,N)>4 THEN Y(4)=N
2710 NEXT N
2720 IF Y(1)-Y(2)=0 THEN V=3.141593/2:GOTO 2740
2730 V=ATN(10/(Y(1)-Y(2)))
2740 D=V*(180/3.141593)
2750 IF D>0 THEN D=-1*(90-D):GOTO 2770
2760 IF D<0 THEN D=90-ABS(D)
2770 LOCATE 7,40:PRINT "Angle of rotation ";D
2780 L=((Y(3)+Y(4))/2)
2790 X=(L-8)
2800 LOCATE 9,40:PRINT "Center offset ";X*.071;" inches"
2810 LOCATE 22,2:PRINT "return to continue"
2820 IF INKEY$="" THEN GOTO 2820
2830 RETURN
2840 :
2850 REM ***** get info from disk and display it *****
2860 OPEN "I",#3,"lordinfo.doc"
2870 N=0
2880 CLS
2890 WHILE N<21

```

```
2900 IF EOF(3) THEN GOTO 2990
2910 LINE INPUT#3,A$
2920 PRINT A$
2930 N=N+1
2940 WEND
2950 N=0
2960 PRINT:PRINT "press any key to cont"
2970 IF INKEY$="" THEN GOTO 2970
2980 GOTO 2880
2990 PRINT:PRINT "press any key to return to menu"
3000 IF INKEY$="" THEN GOTO 3000
3010 CLOSE #3
3020 RETURN
3030 :
3040 REM ***** ERROR TRAPPING *****
3050 IF ERR=57 AND ERL=1250 THEN CLOSE:RESUME 1230
3060 IF ERR=57 AND ERL=1270 THEN RESUME 1270
3070 PRINT ERR,ERL:BEEP
3080 RESUME
3090 DATA 0,8,7,1,9,4,12,6,5,13,2,10,3,14,11,15
```

Autoexec.bat File

```
Echo off
cls
graphics
echo /-----/
echo /          Program for the Lord Tactile sensor          /
echo /                                                                 /
echo /          Minimum Hardware requirements                  /
echo /                      RGB monitor                        /
echo /                      Com1: installed                    /
echo /                                                                 /
echo /          Make Sure the Sensor is ON                    /
echo /-----/
pause
lordcom
```